Math 277
Ordinary Differential Equations

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 12
Systems of Differential Equations

Initializations

```
> restart;
```

12.1 Systems of Differential Equations with Constant Coefficients

Examples

Example 12.1.1

Solve the initial value problem

$$\frac{dx}{dt} = -7\,x + 10\,y + t$$
$$\frac{dy}{dt} = -3\,x + 4\cdot y + 1$$
$$x(0) = 5,\ y(0) = -4$$

Solution 1

There are a variety of ways in which a system like this can be solved. We start with the most intuitive. Solve the first equation for $y(t)$ and substitute the result into the second equation. This produces a second order differential equation for $x(t)$.

```
> deq1:=diff(x(t), t)+7*x(t)-10*y(t)-t=0;
  deq2:=diff(y(t), t)+3*x(t)-4*y(t)-1=0;
```

$$deq1 := \frac{d}{dt}\,x(t) + 7\,x(t) - 10\,y(t) - t = 0$$

$$deq2 := \frac{d}{dt}\,y(t) + 3\,x(t) - 4\,y(t) - 1 = 0 \qquad \textbf{(2.1.1.1)}$$

```
> y_x:=isolate(deq1, y(t));
  deq3:=eval(deq2, y_x);
```

$$y\_x := y(t) = \frac{1}{10}\,\frac{d}{dt}\,x(t) + \frac{7}{10}\,x(t) - \frac{1}{10}\,t$$

$$deq3 := \frac{1}{10}\,\frac{d^2}{dt^2}\,x(t) + \frac{3}{10}\,\frac{d}{dt}\,x(t) - \frac{11}{10} + \frac{1}{5}\,x(t) + \frac{2}{5}\,t = 0 \qquad \textbf{(2.1.1.2)}$$

A little bit of handy work can simplify the appearance of this expression.

```
> deq4:=10*deq3;
```

(2.1.1.3)

$$deq4 := \frac{d^2}{dt^2} x(t) + 3 \left(\frac{d}{dt} x(t)\right) - 11 + 2x(t) + 4t = 0 \quad \textbf{(2.1.1.3)}$$

```
> deq5:=map(u->u+11-4*t, deq4);
```

$$deq5 := \frac{d^2}{dt^2} x(t) + 3 \left(\frac{d}{dt} x(t)\right) + 2x(t) = 11 - 4t \quad \textbf{(2.1.1.4)}$$

The general solution of this equation is easily found using the method of undetermined coefficients. First generate the solution of the homogeneous equation.

```
> aux:=r^2+3*r+2=0;
  ev:=solve(aux, r);
```

$$aux := r^2 + 3r + 2 = 0$$

$$ev := -1, -2 \quad \textbf{(2.1.1.5)}$$

```
> xgh:=add(c[k]*exp(ev[k]*t), k=1..2);
```

$$xgh := c_1 e^{-t} + c_2 e^{-2t} \quad \textbf{(2.1.1.6)}$$

Find a particular solution of the form $x_p(t) = a_1 + a_2 t$.

```
> xtry:=a[1]+a[2]*t;
  eqa:=eval(deq5, x(t)=xtry);
  vala:=solve(identity(eqa, t), {a[1], a[2]});
  xp:=eval(xtry, vala);
```

$$xtry := a_1 + a_2 t$$

$$eqa := 3a_2 + 2a_1 + 2a_2 t = 11 - 4t$$

$$vala := \left\{a_1 = \frac{17}{2}, a_2 = -2\right\}$$

$$xp := \frac{17}{2} - 2t \quad \textbf{(2.1.1.7)}$$

We conclude that the general solution of the differential equation

$$\frac{d^2}{dt^2} x(t) + 3 \left(\frac{d}{dt} x(t)\right) + 2x(t) = 11 - 4t$$

is given by

$$x_g(t) = c_1 e^{-t} + c_2 e^{-2t} + \frac{17}{2} - 2t$$

```
> xg:=xgh+xp;
```

$$xg := c_1 e^{-t} + c_2 e^{-2t} + \frac{17}{2} - 2t \quad \textbf{(2.1.1.8)}$$

The **corresponding** $y_g(t)$ is obtained by substitution of $x(t) = x_g(t)$ in the formula

$$y(t) = \frac{1}{10} \frac{d}{dt} x(t) + \frac{7}{10} x(t) - \frac{1}{10} t$$

```
> yg:=eval(rhs(y_x), x(t)=xg);
```

$$yg := \frac{3}{5} c_1 e^{-t} + \frac{1}{2} c_2 e^{-2t} + \frac{23}{4} - \frac{3}{2} t \quad \textbf{(2.1.1.9)}$$

Finally, we implement the initial conditions $x(0) = 5$, $y(0) = -4$ and solve for the constants c_1 and c_2.

```
> eq1:=eval(xg, t=0)=5;
  eq2:=eval(yg, t=0)=-4;
  valc:=solve({eq1, eq2}, {c[1], c[2]});
```

$$eq1 := c_1 + c_2 + \frac{17}{2} = 5$$

$$eq2 := \frac{3}{5} c_1 + \frac{1}{2} c_2 + \frac{23}{4} = -4$$

$$valc := \left\{ c_1 = -80, c_2 = \frac{153}{2} \right\} \quad \mathbf{(2.1.1.10)}$$

We conclude that the solution of the given initial value problem is given by

```
> x(t)=eval(xg, valc);
  y(t)=eval(yg, valc);
```

$$x(t) = -80\, e^{-t} + \frac{153}{2} e^{-2t} + \frac{17}{2} - 2t$$

$$y(t) = -48\, e^{-t} + \frac{153}{4} e^{-2t} + \frac{23}{4} - \frac{3}{2} t \quad \mathbf{(2.1.1.11)}$$

Solution 2
We can write the system

$$\frac{dx}{dt} = -7x + 10y + t$$
$$\frac{dy}{dt} = -3x + 4\cdot y + 1$$

in operator notation

$$(\mathrm{D} + 7)[x] - 10[y] = t$$
$$3\,[x] + (\mathrm{D}-4)[y] = 1$$

and then solve this system by elimination much like we would solve an ordinary linear system in two unknowns.

```
> deq1:=diff(x(t), t)+7*x(t)-10*y(t)=t;
  deq2:=3*x(t)+diff(y(t), t)-4*y(t)=1;
```

$$deq1 := \frac{\mathrm{d}}{\mathrm{d}t} x(t) + 7\,x(t) - 10\,y(t) = t$$

$$deq2 := 3\,x(t) + \frac{\mathrm{d}}{\mathrm{d}t} y(t) - 4\,y(t) = 1 \quad \mathbf{(2.1.1.12)}$$

First eliminate the variable x.

```
> deq3:=3*deq1;
  deq4:=diff(deq2, t)+7*deq2;
```

$$deq3 := 3 \left(\frac{\mathrm{d}}{\mathrm{d}t} x(t) \right) + 21\,x(t) - 30\,y(t) = 3\,t$$

$$deq4 := 3 \left(\frac{\mathrm{d}}{\mathrm{d}t} x(t) \right) + \frac{\mathrm{d}^2}{\mathrm{d}t^2} y(t) + 3 \left(\frac{\mathrm{d}}{\mathrm{d}t} y(t) \right) + 21\,x(t) - 28\,y(t) = 7 \quad \mathbf{(2.1.1.13)}$$

```
> deq5:=deq3-deq4;
```

$$deq5 := -2\,y(t) - \left(\frac{d^2}{dt^2}\,y(t)\right) - 3\left(\frac{d}{dt}\,y(t)\right) = 3\,t - 7 \qquad \textbf{(2.1.1.14)}$$

Solve this differential equation using the method of undetermined coefficients. You know how to do that, right! For convenience we will just use the **dsolve** command.

```
> soly:=dsolve(deq5, y(t));
```

$$soly := y(t) = \frac{23}{4} - \frac{3}{2}\,t - e^{-2\,t}\,\_C1 + e^{-t}\,\_C2 \qquad \textbf{(2.1.1.15)}$$

Let us replace the system generated variables $\_C1$ and $\_C2$ by b_1 and b_2. (This has more than just an esthetic value)

```
> soly:=eval(soly, {seq(_C||k=b[k], k=1..2)});
```

$$soly := y(t) = \frac{23}{4} - \frac{3}{2}\,t - e^{-2\,t}\,b_1 + e^{-t}\,b_2 \qquad \textbf{(2.1.1.16)}$$

Analogous to what we did in our first solution, we could substitute this into the second differential equation

$$3\,x(t) + \frac{d}{dt}\,y(t) - 4\,y(t) = 1$$

and solve directly for $x(t)$. Alternatively however, we can eliminate the variable $y(t)$ and find $x(t)$ that way.

```
> deq1;
  deq2;
```

$$\frac{d}{dt}\,x(t) + 7\,x(t) - 10\,y(t) = t$$

$$3\,x(t) + \frac{d}{dt}\,y(t) - 4\,y(t) = 1 \qquad \textbf{(2.1.1.17)}$$

```
> deq6:=diff(deq1, t)-4*deq1;
  deq7:=-10*deq2;
```

$$deq6 := \frac{d^2}{dt^2}\,x(t) + 3\left(\frac{d}{dt}\,x(t)\right) - 10\left(\frac{d}{dt}\,y(t)\right) - 28\,x(t) + 40\,y(t) = 1 - 4\,t$$

$$deq7 := -30\,x(t) - 10\left(\frac{d}{dt}\,y(t)\right) + 40\,y(t) = -10 \qquad \textbf{(2.1.1.18)}$$

```
> deq8:=deq6-deq7;
```

$$deq8 := \frac{d^2}{dt^2}\,x(t) + 3\left(\frac{d}{dt}\,x(t)\right) + 2\,x(t) = 11 - 4\,t \qquad \textbf{(2.1.1.19)}$$

This differential equation can be solved using the method of undetermined coefficients. For expedience we once again use the **dsolve** routine.

```
> solx:=dsolve(deq8, x(t));
  solx:=eval(solx, {seq(_C||k=a[k], k=1..2)});
```

$$solx := x(t) = \frac{17}{2} - 2\,t - e^{-2\,t}\,\_C1 + e^{-t}\,\_C2$$

$$solx := x(t) = \frac{17}{2} - 2\,t - e^{-2\,t}\,a_1 + e^{-t}\,a_2 \qquad \textbf{(2.1.1.20)}$$

Now it is important to realize that the parameters a_1, a_2, b_1, and b_2 are not independent. Substitution of

$$x(t) = \frac{17}{2} - 2\,t - e^{-2\,t}\,a_1 + e^{-t}\,a_2$$

and

$$y(t) = \frac{23}{4} - \frac{3}{2}\,t - e^{-2\,t}\,b_1 + e^{-t}\,b_2$$

into the first differential equation

$$\frac{d}{dt}\,x(t) + 7\,x(t) - 10\,y(t) = t$$

yields

```
> eq:=eval(deq1, {solx, soly});
```

$$eq := -5\,e^{-2\,t}\,a_1 + 6\,e^{-t}\,a_2 + t + 10\,e^{-2\,t}\,b_1 - 10\,e^{-t}\,b_2 = t \qquad \textbf{(2.1.1.21)}$$

and because the functions $e^{-2\,t}$ and e^{-t} are linearly independent, this yields

```
> parsb:=solve(identity(eq, t), {b[1], b[2]});
```

$$parsb := \left\{ b_1 = \frac{1}{2}\,a_1,\, b_2 = \frac{3}{5}\,a_2 \right\} \qquad \textbf{(2.1.1.22)}$$

Therefore, the general solution of the system is given by

$$x(t) = \frac{17}{2} - 2\,t - e^{-2\,t}\,a_1 + e^{-t}\,a_2$$

$$y(t) = \frac{23}{4} - \frac{3}{2}\,t - \frac{1}{2}\,e^{-2\,t}\,a_1 + \frac{3}{5}\,e^{-t}\,a_2$$

```
> xg:=rhs(solx);
  yg:=eval(rhs(soly), parsb);
```

$$xg := \frac{17}{2} - 2\,t - e^{-2\,t}\,a_1 + e^{-t}\,a_2$$

$$yg := \frac{23}{4} - \frac{3}{2}\,t - \frac{1}{2}\,e^{-2\,t}\,a_1 + \frac{3}{5}\,e^{-t}\,a_2 \qquad \textbf{(2.1.1.23)}$$

Finally we again implement the initial conditions and arrive at the desired solution of the initial value problem.

```
> eq1:=eval(xg, t=0)=5;
  eq2:=eval(yg, t=0)=-4;
  vala:=solve({eq1, eq2}, {a[1], a[2]});
```

$$eq1 := \frac{17}{2} - a_1 + a_2 = 5$$

$$eq2 := \frac{23}{4} - \frac{1}{2}\,a_1 + \frac{3}{5}\,a_2 = -4$$

$$vala := \left\{ a_1 = -\frac{153}{2},\, a_2 = -80 \right\} \qquad \textbf{(2.1.1.24)}$$

We conclude that the solution of the given initial value problem is given by

```
> x(t)=eval(xg, vala);
  y(t)=eval(yg, vala);
```

$$x(t) = -80\,e^{-t} + \frac{153}{2}\,e^{-2t} + \frac{17}{2} - 2t$$

$$y(t) = -48\,e^{-t} + \frac{153}{4}\,e^{-2t} + \frac{23}{4} - \frac{3}{2}\,t \qquad \textbf{(2.1.1.25)}$$